URBI Language Specification

v 1.3

©Jean-Christophe Baillie

October 2005

Contents

URE	Language Specification	4
1.1	JRBI Port	4
1.2	Comments	4
1.3	Commands, Time Operators and Messages	4
	.3.1 Time Operators	4
	.3.2 Tagged commands	5
	.3.3 Flags	6
	.3.4 Grouped commands	6
	.3.5 Messages	6
.4	Devices	7
	.4.1 Definition	7
	.4.2 Complementary fields and methods	7
	.4.3 Standard aliases and global device	8
	.4.4 info	8
.5	Variables	9
	.5.1 Variables types	9
	.5.2 Variables syntax	9
	.5.3 Variables declaration	10
	.5.4 Variable properties	11
	.5.5 Variables by name	12
	.5.6 Static variable use	12
	.5.7 Variable aliases	13
	.5.8 Variables inside functions	13
	.5.9 Variable extensions	13
.6	ssignments, Modifiers and blending modes	15
	.6.1 Float assignment	15
	.6.2 String assignment	18
	.6.3 List assignment	19
	.6.4 Binary assignment	19
	.6.5 Blending modes	20
.7	xpressions	20
	.7.1 Float	20
	.7.2 String	21
	.7.3 List	21
	.7.4 Binary	21
	.7.5 Boolean	21
	.7.6 Standard functions	22
.8	Control and event catching structures	22
	.8.1 Control structures	23
	.8.2 Soft tests	25

		1.8.3 Event catching structures	26
	1.9	Emit events	27
		1.9.1 Simple events	27
		1.9.2 Events with parameters	27
		1.9.3 Event duration	28
		1.9.4 Pulsing events: the every command	28
		1.9.5 Predefined pulsing events	28
	1.10	Grouping	28
	1.11	Function definition	29
	1.12	Files	30
	1.13	Standard commands, operators or global variables	30
		1.13.1 echo	30
		1.13.2 exec	31
		1.13.3 stop, block, unblock, freeze, unfreeze	31
		1.13.4 ping	31
		1.13.5 motor on, motor off, motor switch	32
		1.13.6 connections, vars, uservars	32
		1.13.7 stopall, undefall, killall, disconnect	32
		1.13.8 quit, reboot, shutdown, reset	32
		1.13.9 Useful functions that are part of the specification	32
	1.14	URBI Header	34
2	URB	I grammar	35
_	2.1	Bison grammar	
	2.2	Terminal symbols and flex regexp	

Introduction

This document is a full specification of the URBI language. It is not a tutorial, it is a reference manual and a technical specification upon which the URBI Certification will be based. If you are already familiar with URBI, this document may help you get precise information on a given command or help you review more advanced features not covered in tutorials. If you want an introduction to URBI, try the URBI Tutorial on www.urbiforge.com.

Any implementation of an URBI server must comply with the latest version of this specification to be "URBI compliant". It must also pass through the test softwares that will be based on this specification. See the URBI Language License for more information.

The URBI Language copyright holder is currently the only authority which has the ability to deliver a "URBI Compliant" certification.

Chapter 1

URBI Language Specification

URBI (Universal Robotic Body Interface) is a client/server based interpreted language that can be used to control robots or complex systems of any kind. The language defines a standard protocol to give commands and receive messages from the machine to be controlled.

We will describe here in extension the features of the URBI language that must be implemented in any URBI compliant server.

1.1 URBI Port

By default, the URBI Server is listening on port 54000 in a TCP/IP connection, unless stated otherwise. In principle, other ways of sending/receiving commands could be available, like using a shared memory between processes when the client and the server are running on the same host or using something else than TCP/IP. This will depend on the underlying hosting OS and would typically be described in the specific "URBI Doc" manual for your application/robot.

1.2 Comments

In the following, we will use URBI comments to give extra information about code lines given as examples. URBI comments can be specified like in C, C++ with '//' or $'/* \ldots */'$ and also like many scripted languages using ' #', which is equivalent to '//'.

1.3 Commands, Time Operators and Messages

The URBI server receives *commands* from a client and returns *messages* to this client. The normal way of using a URBI controlled robot is to send commands using TCP/IP on the URBI port (54000) and wait for messages in return. A simple telnet client is enough to do that for simple applications (or the more advanced 'URBILab' client), otherwise libraries (liburbi) are available in most programming languages to wrap the TCP/IP sending/receiving job in simple functions.

1.3.1 Time Operators

In URBI, every command has a duration (beside the execution time required by the processor which is assumed to be negligible). Commands are joined by *time operators* like 'semicolon', 'comma',

'pipe' or 'and', which affect the way the commands are serialized or parallelized. We write start(A) the time when command A starts and end(A) the time when it ends. The semantics of the different time operators is the following:

A; B

B starts after A is finished.

$$A; B \Longleftrightarrow start(B) >= end(A)$$

A, B

B starts after A starts.

$$A, B \iff start(B) >= start(A)$$

A&B

B and A start at exactly the same time.

 $A\&B \iff start(B) == start(A)$

Note on zero time execution commands:

Zero time execution commands are commands whose theoretical execution takes zero seconds. Simple assignments, evaluations, etc, are zero time execution commands. Let us consider the three zero time execution commands zero1, zero2 and zero3. In the following case, the order of execution of zero3 compared to zero1 and zero2 is undefined:

```
{zero1;zero2} & zero3
might lead to (1): zero1 zero3 zero2
or: (2): zero1 zero2 zero3
```

The temporal order between &-separated commands is the same as the order of declaration of the commands but the way they are interleaved is undefined by the URBI Specification. In practical situations, and in the current monothreaded implementations of the server that we know, the set of instructions between brackets will be executed first (option 2), but there is no guarantee.

A|B

B starts immediately after A is finished.

 $A|B \iff start(B) == end(A)$

1.3.2 Tagged commands

Any URBI command can be prefixed by a *tag* followed by a colon. If no tag is specified, notag is assumed to be the default tag.

Examples:

```
x = 12; // notag
my_tag: y = x+12; // the tag is 'my_tag'
```

1.3.3 Flags

+C

The tag can be followed by *flags* to ask for meta information/action about the command execution. Any information is given via system messages (prefixed by ***). You might use flags without a tag. The available flags are:

+begin	:	The system message "*** begin" will be displayed when the command starts.
+end	:	The system message "*** end" will be displayed when the command ends.
+report	:	This has the combined effect of +begin and +end.
+error	:	Notify of any error during the command execution.
connection(s)	:	Run the command in the connection whose ID is given by the string s. If s is "all", the command will be run in every connections. If s is "other", it will be run in every connections except the one from which the command is executed. Beware that no context element from your connection (like variables) are transmitted. You actually run the command in the other connection, not in yours.

+bg : Run the command in background.

Examples (see below for explanations on system messages):

```
mytag +report: a = 4;
[136901543:mytag] *** begin
[136901543:mytag] *** end
```

1.3.4 Grouped commands

Commands can be grouped between brackets like in the following example:

{ x=4; y=5, x=x*7 & {y=x-1;w=y+1} | z=y };

The group of commands between brackets is considered as a command itself and therefore must be separated from further commands with a time operator (it can also be tagged, stopped, freezed or blocked like any other command). The last command between brackets does not need a time operator. The bracket-command starts when the first command inside starts and stops when all commands inside have stopped.

1.3.5 Messages

When a command returns a value or when it fails, the URBI server returns a message. The format of a message is the following:

[timestamp:tag] message

The time stamp is the uptime of the server in milliseconds when the message has been sent. The tag is the tag of the associated command or notag if there was no tag specified. The message can be a value (float, string, list or binary) or a system message prefixed by three stars: *******. Error messages or information messages are system messages.

Here is a typical example of commands with their messages in return:

```
1+1;
[136901543:notag] 2.0000000
my_tag:6*6;
[136904711:my_tag] 36.0000000
impossible:1/0;
[136471768:impossible] *** Division by zero
[136471768:impossible] *** EXPR evaluation failed
```

The ability to tag commands is a key feature of URBI since it allows the client to retrieve the results of a specific command in a flow of messages.

1.4 Devices

1.4.1 Definition

Every sensor, motor or controllable element of the robot is a *device*. It has a device name and can be entirely accessed using this name as prefix. With any URBI server there should be available an extensive list of the existing devices on that particular robot, in the "URBI Doc" document.

Every device is associated with a set of *fields* and *methods* that can be accessed using the syntax device.field or device.method(...). Fields are device specific variables and method are device specific functions.

Examples:

Let us consider the device headPan associated with the horizontal motor of the Aibo robot's head. The value of the motor angle can be specified using the val field:

headPan.val = 45;

This will set the headPan motor to 45 degrees, using the variable headPan.val. Note that, for convenience, the val field is directly accessible by default when no field is specified, because headPan is recognized as a known device. So, the previous command can be written:

headPan = 45;

Reading the position of the motor is simply done by evaluating the value of the headPan.val variable:

```
headPan.val; // or simply: headPan;
[178101888:notag] 45.0178740611
```

1.4.2 Complementary fields and methods

Other devices like cameras typically have several fields accessible, like in the case of Aibo (see the URBI Doc of Aibo for more details):

```
camera.resolution = 1;
camera.format = 0;
camera.width;
[188141887:notag] 104.000000
In the case of Aibo, the speaker device (commanding the robot speaker) has a play
```

method that plays a sound given as a way file on the memorystick:

speaker.play("test.wav");

Detailed documentation on what fields and methods are available on a particular robot device should be available in the robot's "URBI Doc" file.

However, some fields are mandatory for any device in a URBI compliant server. The val field is mandatory and, depending of the underlying hardware, the force field might be requested for a motor device. For devices relative to joints or motor, the 'load' field is also mandatory. It controls the load of the joint: 0 means that the joint is loose, 1 means that the joint is not movable by hand (any applied torque will be exactly compensated). Any value in between gives an intermediary situation.

1.4.3 Standard aliases and global device

The variable global.nbdevices returns the number of available devices in the system. You have access to convenient aliases to each device value with global.device[i], pointing to the *i*th device.val variable.

The variable global.devicename [i] returns a string with the name of the *i*th device.

1.4.4 info

Using the command info device gives detailed and useful information on the device in one shot:

```
info headPan;
[01544390:notag] *** device description: Head pan
[01544390:notag] *** device name: headPan
[01544390:notag] *** current value: 73.912655
[01544390:notag] *** current device load: 1.000000
[01544390:notag] *** rangemin: -91.000000
[01544390:notag] *** rangemax: 91.000000
[01544390:notag] *** speedmin: 0.000000
[01544390:notag] *** speedmax: +INF
[01544390:notag] *** unit: deg
```

The info command can also be used with any variable (see "variables" in next section). If the variable belongs to a device, the device name and description and load are also given:

```
info myvar;
[01544390:notag] *** current value: 10.000000
[01544390:notag] *** rangemin: -INF
[01544390:notag] *** rangemax: +INF
[01544390:notag] *** speedmin: 0.000000
[01544390:notag] *** speedmax: +INF
[01544390:notag] *** unit: unspecified
info headPan.x;
[01544390:notag] *** device description: Head pan
[01544390:notag] *** device name: headPan
[01544390:notag] *** current value: 11.000000
[01544390:notag] *** current device load: 1.000000
[01544390:notag] *** rangemin: -INF
[01544390:notag] *** rangemax: +INF
[01544390:notag] *** speedmin: 0.000000
```

```
[01544390:notag] *** speedmax: +INF
[01544390:notag] *** unit: unspecified
```

1.5 Variables

1.5.1 Variables types

Variable can have the following types:

- float: expressed with a point
- time: these are in fact floats conversion of expressions like 3h45m12s into milliseconds. Available time symbols are: d, h, m, s, ms. They can be composed at will. The main usage is to facilitate expression of time values in complex commands.
- string: between double quotes and with a C like syntax for quoted symbols
- list: comma separated and between square brackets. Can be heterogeneous, like [1.2, 3, "hello"]
- binary

Note that there is no real integer or boolean type in URBI. When an integer value is expected, the float is automatically casted into an int. Binary variables are specified by the keyword BIN followed by the size the binary data and a set of freely attributed parameters useful to characterize the binary. A typical use of binary variables is to store sound like wav data, or images as jpeg.

1.5.2 Variables syntax

Variable names are composed of a prefix and a suffix separated by a point. When there is no prefix, the variable is assumed to be local to the connection or to the function call (a virtual prefix equal to the connection or function call identifier is silently created). Variable can also be indexed or multiindexed by integer values (float are rounded to integers) or strings between brackets, allowing to use flexible arrays in URBI.

Examples:

Here are a few examples of variables assignments:

```
i = 4;
j = i;
s = "hello";
s2 = s + " world!";
mylist=[1,2,"hello",4];
myemptylist=[];
```

1.5.3 Variables declaration

The declaration of a variable is implicit in URBI and is done via any assignment. Variable used without having been assigned a value first will be rejected as unknown. Variable types are inferred automatically during the first assignment.

def, defcheckon, defcheckoff

Automatic and implicit variable declaration is a convenient feature for fast scripting but can be the source of bugs for larger programs. The main problem arises from mistyped variable names which can go unnoticed in an assignment.

To provide with a way to allow name-safe coding in URBI, you can explicitly define variables by hand with the def keyword and activate the variable definition checking with the command defcheckon.

```
a = 0; // that's ok
defcheckon;
b = 0; // b is not implicitly defined
[00025254:notag] *** Unknown identifier: U597753864.b
def b = 0; // now it's ok again
```

The syntax of def is very flexible. Like in the example above, you can perform an assignment prefixed by def to make sure your variable is defined, but you can also define variables independently of assignments, for example at the beginning of your program. In particular, you can define several variables at a time like this:

```
def a;
def b;
def stuff.d; <=>
def {
    a;
    b;
    stuff.d;
};
```

Or define several variables with the same prefix like this:

```
def stuff.a;
def stuff.b;
def stuff.d; <=>
def stuff {
```

a; b; d; };

The variable definition checking can be turned off with the command defcheckoff. Note that when the definition checking is on, type mismatches are signaled when they occur.

undef, undefall, isdef

To release a variable (and the associated memory), the undef operator is available:

```
a = 0;
a = "hi";
[00025254:notag] *** U597753864.a type mismatch
undef a;
a = "now it's ok";
showit:a;
[00071210:showit] "now it's ok"
```

You can release all user defined variables with the command undefall.

The isdef(x) function returns 1 when x is defined or zero otherwise.

1.5.4 Variable properties

One originality of URBI is that any variable is associated to a set of properties. For example, there is a rangemin and rangemax property available. The property can be read/set via an indirection -> on the variable name:

```
x = 0;
x->rangemax = 10;
x->rangemax;
[00035124:notag] 10.000000
legRF1.val->rangemin; // or simply: legRF1->rangemin
[00035834:notag] -134.000000
```

The following properties are available:

```
rangemin: the minimal value accepted. Outranging values are clipped. default = -inf
rangemax: the maximal value accepted. Outranging values are clipped. default = +inf
speedmax: the maximal speed (units/s) for the evolution of the variable. default = +inf
speedmin: the minimal speed (units/s) for the evolution of the variable in a speed mod-
ified assignment. default = 0
unit: variable unit. default = "", usual values include "def", "cm", "m",...
delta: variable tolerance used with the adaptive modifier (explained in 1.6.1)
blend: variable blend mode (explained in 1.6.5)
```

1.5.5 Variables by name

The syntaxic construction s(s) where s is a string is equivalent to the variable whose name is given by s. For example, the two following constructions are equivalent:

```
$("global.val") <=> global.val
```

Symmetrically, the operator % applied on a variable name gives the string equal to the name of the variable:

```
%global.val;
[01544390:notag] "global.val"
```

Using these features, the name of a variable can be used as a reference when passing function parameters for example. The string type plays the role of a pointer in that case.

1.5.6 Static variable use

In some cases, it is necessary to have a variable in an expression that will be evaluated only once the first time the expression is calculated and which will keep this original value if the expression is re-evaluated later. This is called "static variable use".

To request a static evaluation of a variable, the name of the variable must be prefixed by @

The following example shows how the static declaration might affect the behavior of a wait command whose test is based on the value of a variable x (see commands description for the semantics of "wait"):

```
x=0;
{wait (x==1) | ping},
x=1;
[00943009:notag] *** pong time=943010.110000
x=0;
{wait (@x==1) | ping},
x=1;
// nothing happens, the value of x is still 0 in the test evaluation.
```

This feature is useful with loop indexes used in modifiers (see below). Since the index will change while the loop is executed, it is necessary that the expression evaluation keeps the original value of the index instead of using the current value (which at the end of the loop will be identical for all instances). Example:

```
for & (i=0; i<5;i++) {
    x = 0 time: (100 * @i) |
    echo "timer is over"
}</pre>
```

Variables inside array indexes are automatically made static for convenience, since there would be no point otherwise. So, tab[i] and tab[:i] are identical.

1.5.7 Variable aliases

The alias command can be used to create variable aliases. It can be used for example to rename variables for compatibility or to conveniently access a set of variables using an array and a numerical index.

The alias command without parameters shows the list of existing aliases.

When they are used, aliases are chained until they reach a variable for which no alias exists.

```
alias here.a there.b;
alias here.tab[1] there.b1;
alias here.tab[2] there.b2;
alias;
[00943009:notag] *** here.a -> there.b
[00943009:notag] *** here.tab_1 -> there.b1
[00943009:notag] *** here.tab_2 -> there.b2
```

The second parameter of an alias command is chained, in case there is already an alias defined for it, but the first parameter is not.

1.5.8 Variables inside functions

Variables without a prefix inside a function are local to the function call. This is essential for recursive calls. You can access local variables belonging to your connection from inside a function by using the prefix local which will be transformed into the connection prefix. Consider the following example:

1.5.9 Variable extensions

Variables names can be followed by several "extensions" which will modify the way the variable evaluation or assignment is interpreted. Extensions are defined by an apostrophe, followed by one or several characters describing the extension:

'n : normalization

Using the rangemin and rangemax properties, it is possible to calculate a normalized value between 0 and 1 for any variable. This is done with the 'n extension. This can be useful to achieve a limited sort of robot independent commands. Example:

```
x = 4;
x->rangemin = 0;
x->rangemax = 20;
x;
[00943009:notag] 4.000000
x'n;
```

```
[00943009:notag] 0.200000
x'n = 0.5;
x;
[00943009:notag] 10.000000
```

If rangemin or rangemax are not defined, URBI is unable to calculate the normalized value and complains about it:

```
y = 4;
y'n;
[00943009:notag] !!! Impossible to normalize: no range defined for variable...
[00943009:notag] !!! EXPR evaluation failed
```

' : first order theoretical derivative

This extension returns the first order derivative of the variable assignments. When used on a motor device value for example, it returns the theoretical derivative of the input command on the motor (not the actual real motor position derivative as it could be estimated from the sensor output).

```
x = 0;
x->blend = add; // add mode to aggregate assignments
x = 100 speed:33 &
x = 400 speed:12, // the derivative will be the added speeds
x';
[00143009:notag] 45.000000
```

" : second order theoretical derivative

This extension returns the second order derivative of the variable assignments. Like the first order derivative above, it reflects the theoretical assignments and not the real variable change in case of a motor value.

```
x = 0;
x->blend = add; // add mode to aggregate assignments
x = 100 accel:3 &
x = 400 accel:2.5, // the derivative will be the added speeds
x'';
[00143009:notag] 5.500000
```

'd : first order real derivative

This extension returns the first order derivative of the variable value. In most cases, it is exactly the same as the theoretical first order derivative described above. In the case of motor values or sensors however, the theoretical value (coming from an assignment command) might differ from the real value taken by the motor and reported by the sensor. This extension measure this real value derivative.

```
motoroff;
// show 10s of the headPan motor value "sensed" derivative
+timeout(10s):loop headPan'd;
```

' dd : second order real derivative

This extension returns the second order derivative of the variable value and is described like the first order real derivative described above.

'e : "theoretical vs real" error estimation

This extension returns the difference between the theoretical value that the variable should have according to its underlying assignments and the real value coming from sensors. If the variable is not a motor or a sensor, this error is always zero. Otherwise, in the case of motors, it measures the latency in position which often reflects the force currently applied on the motor (the greater the difference, the greater the "effort" to compensate it).

In practical situations, this quantity can be used to measure the force on a motor or to detect potential jams (if the motor is stopped in the course of a trajectory, the error absolute value will grow).

1.6 Assignments, Modifiers and blending modes

Assignments in URBI are of the form:

variable = value [modifiers]

Modifiers are optional parameters that affect the semantics of the assignment. They are described as a list of couples modifier_name:value after the main assignment value. *The order of the modifiers in the list has no influence*.

There are three types of assignments, depending on the type of the value:

1.6.1 Float assignment

We describe here the available modifiers for a float assignment. Modifiers for float assignments allows the float value to be assigned with different speed, time limit, motion profile... To use a modifier, the variable that is assigned must have an *initial value*, otherwise a "no start value" error is generated (except for the sin modifier). The value to be assigned is called the *target value*.

It is possible to limit the speed at which values are reached using the speedmin and speedmax properties. Setting speedmax to *s* (units/seconds) insures that the corresponding variable will never move faster than *s*. speedmin is taken into account by certain commands only (see below), to insure that the device will move of a minimal significant amount at each time step.

time:t

time:t means that the target value must be reached in t milliseconds, starting from the initial value (t can be expressed as a time value, like 3m2lsl5ms). A linear trajectory over time is computed and executed. The assignment command terminates when the time is over. The trajectory is not recalculated if the value deviate from the initial trajectory planned. URBI tries to stick back to the pre-calculated trajectory when it can and if there is enough time left.

Example:

x = 0; x = 100 time:1s; // reaches value 100 in 1 second.

Note that there is no guarantee that the target value will be reached. It is reached only if the time is enough for the system to reach it, taken into account the fact that the speed of some variables linked to motors is limited by the hardware. It will also be reached if there is no other assignment command acting on the corresponding variable just after this command.

time:t adaptive:1

time: t means that the target value must be reached in t milliseconds, starting from the initial value. The *adaptive* modifier set to 1 indicates that the trajectory should not be precalculated but adjusted at each time step. So if the value deviates from the normal trajectory, the command will adapt itself. The command terminates when the target value is actually reached.

Example:

If this command is used with a motor, it is possible to set a minimum speed, using the speedmin field of the corresponding device. The reason of this feature is that when the speed is too low, the joint will never move and, since the trajectory is recalculated at each time step, it will not evolve properly (the initial value will remain always the same).

The delta property of the variable is used to set a tolerance in the definition of "the target value is reached". To be precise, the target value is reached +/- variable->delta. The delta property should reflect the hardware precision. Setting delta too low could result in the target never being reached because the reaching cannot be detected. Theoretical variables (not related to a device) have a value of zero for delta by default.

smooth:t

moth:t means that the target value must be reached smoothly in t milliseconds, starting from the initial value. A smooth trajectory with a sinusoidal profile is computed and executed. The assignment command terminates when the time is over. The trajectory is not recalculated if the value deviate from the initial trajectory planned. URBI tries to stick back to the pre-calculated trajectory when it can and if there is enough time left.

Example:

Note that there is no guarantee that the target value will be reached. It is reached only if the time is enough for the system to reach it, taken into account the fact that the speed of some variables linked to motors is limited by the hardware. It will also be reached if there is no other assignment command acting on the corresponding variable just after this command.

speed: s means that the target value must be reached with a constant speed of *s* units per seconds. A linear trajectory over time is computed and executed. The assignment command terminates when the corresponding time, calcutated from the speed, is over. The trajectory is not recalculated if the value deviate from the initial trajectory planned. URBI tries to stick back to the pre-calculated trajectory when it can and if there is enough time left.

Example:

```
x = 0;
x = 100 speed:10; // reaches 100 with a speed of 10 units/sec
```

Note that there is no guarantee that the target value will be reached. It is reached only if the time is enough for the system to reach it, taken into account the fact that the speed of some variables linked to motors is limited by the hardware. It will also be reached if there is no other assignment command acting on the corresponding variable just after this command.

speed:s adaptive:1

speed: s means that the target value must be reached with a constant speed of *s* units per seconds. The *adaptive* modifier set to 1 indicates that the trajectory should not be precalculated but adjusted at each time step. So if the value deviates from the normal trajectory, the command will adapt itself. The command terminates when the target value is actually reached.

Example:

If this command is used with a motor, it is possible to set a minimum speed, using the speedmin field of the corresponding device. The reason of this feature is that when the speed is too low, the joint will never move and, since the trajectory is recalculated at each time step, it will not evolve properly and might even not move at all (the initial value will remain always the same).

The delta property of the variable is used to set a tolerance in the definition of "the target value is reached". To be precise, the target value is reached +/- variable->delta. The delta property should reflect the hardware precision. Setting delta too low could result in the target never being reached because the reaching cannot be detected. Theoretical variables (not related to a device) have a value of zero for delta by default.

accel:a

accel: a means that the target value must be reached with a constant acceleration of *a* units per seconds, with an initial speed of zero. A parabolic trajectory is computed and executed. The assignment command terminates when the time corresponding to the execution of the trajectory is over. The trajectory is not recalculated if the value deviate from the initial trajectory planned. URBI tries to stick back to the pre-calculated trajectory when it can and if there is enough time left.

Example:

Note that there is no guarantee that the target value will be reached. It is reached only if the time is enough for the system to reach it, taken into account the fact that the speed of some variables linked to motors is limited by the hardware. It will also be reached if there is no other assignment command acting on the corresponding variable just after this command.

sin:t ampli:a phase:p getphase:q

ampli, getphase and phase are optional modifiers to be used with the sin modifier. Default values are zero for ampli and phase.

sin:t ampli:a phase:p getphase:q means that the value will oscillate in a sinusoidal manner around the target value, with a period of t milliseconds, an amplitude of a units and a phase of p radians. The current value of the phase will constantly be stored in the q variable (q must be a valid variable name). The modifier cos can be used instead of sin to set the default phase to $\pi/2$. The corresponding trajectory is executed in loop and *the command never terminates*.

The trajectory is not recalculated if the value deviate from the initial trajectory planned. URBI always tries to stick back to the pre-calculated trajectory.

Example:

```
headPan'n = 0.5 sin:2s ampli:0.5,
/* since the command does not terminate, it is better
   to end it with a comma. This command will oscillate
   the head will full amplitude in a 2s periodic movement.
*/
```

Note that by using a conjunction of phase and getphase on the same variable, you can store the value of the phase on exit and restart where the command was if the assignment is rerun:

headPan'n = 0.5 sin:2s ampli:0.5 phase:a getphase:a,

timeout:t

timeout:t means that the assignment command will terminate after t milliseconds, regardless of the motion profile set by other modifiers. This can be especially useful to insure a command will terminate or in conjunction with a sin modifier to implicitly set a limited number of oscillation periods.

It is recommended to use a +timeout flag instead. This modifier is kept for compatibility with previous versions.

1.6.2 String assignment

Strings can be assigned like floats, and they are delimited by quotation marks.

Example:

s = "hello";

Modifiers can be used to set local variables used with a dollar prefix inside the string, and compose elaborated strings, a bit like printf. The following example illustrates the syntax:

```
s = "hello $name. $name's age is $age" name:"John" age:34;
s;
[01544390:notag] "hello John. John's age is 34.000000"
```

1.6.3 List assignment

Lists can be assigned like floats or string, and they are delimited by square brackets.

Example:

```
l = [11, "hello", 7.1];
q = [];
m = [1,2] + [3,5];
```

You can add lists together or with new members. Lists can be heterogeneous. See the foreach command to learn how to scan lists.

1.6.4 Binary assignment

A binary assignment is a very special command in URBI (it has a specific definition in the grammar of the language). It can not be inserted in a flow of commands between brackets or separated by a "comma", a "pipe" or a "and". It must be a single command, terminated by a semicolon. The syntax is:

```
variable = bin <size> <list of parameters>;
```

The size is the number of bytes in the binary variable. The list of parameters is a freely attributed list of float values or identifiers, describing meta information about the binary buffer. This is used for example to specify the sound format in a wav binary. The convention is that the first of these parameters is the format of the binary data, like for example jpeg, raw, wav, mp3, YCbCr,

Just after the ending semicolon comes the binary information. After the specified number of bytes, the URBI parser switch back to ascii mode and waits for another command. The switching between ascii and binary is the reason why binary assignments are special commands handled differently by the parser.

This is an example that plays a sound with Aibo (the server automatically converts the sound format, using the parameters):

Hard copies

It is also possible to assign a preexisting binary variable to another variable, like in newbin = oldbin;. However, the copy is a pointer copy, to save memory. So, in the given example above, everything that is done to oldbin is done to newbin too. To make a hard copy of a binary variable, use the copy operator:

1.6.5 Blending modes

Blending modes is an important feature of URBI. Since the server is a multiclient server, it is possible that two clients (or even the same client) start conflicting assignment commands for the same variable at the same time. To handle this problem, URBI defines six blending modes:

- normal : This is the default mode for a new variable. When several conflicting assignments run at the same time, the last one to execute sets the value. The others run silently in the background, eventually coming back in front if the last terminates.
 - mix : In this mode, conflicting commands are mixed and averaged. If one command is to increase with a speed of 10 and the other command is to increase with a speed of 4, the result will be an increase of speed 7.
 - add : In this mode, conflicting commands are added. If one command is to increase with a speed of 10 and the other command is to increase with a speed of 3, the result will be an increase of speed 13.
 - queue : Conflicting commands are queued and executed one after the other.
- discard : Any conflicting command is ignored and suppressed.
- cancel : Any new command that is conflicting with previously existing and running commands will terminate these commands and take the place.

The default mode is normal, with one exception: sound binary variables (like the Aibo speaker.val) should be generally set to the queue mode by default. The effect will be to queue the sounds in the internal sound buffer of the robot. For those sound binary variables, mix or add acts like a sound mixer. The other modes have an obvious meaning.

Blending modes on variables can be checked by evaluation the blend property. It returns a string corresponding to the current blending mode of the variable:

```
foe->blend;
[01544390:notag] "normal"
speaker.val->blend; // or simply: speaker->blend;
[01544390:notag] "gueue"
```

Blend mode constants are defined in URBI to set the blend mode easily. These constants are: add, mix, queue, discard, cancel, normal. They are simple shortcuts to the corresponding string.

```
foe->blend = cancel;
speaker->blend = mix;
foe->blend = "add"; // works also with the explicit string
```

1.7 Expressions

The different existing types (float, string, list or binary) can be combined with different operators.

1.7.1 Float

The standard + - / \star \land operators are available. The underlying type is the C++ "double".

It is possible to change the displayed precision of floats by changing the global.nbdigits variable. Default value is 6. However, the underlying calculus will still perform with the maximal decimal precision allowed on the robot hardware, independently of the number of digits displayed.

1.7.2 String

For string values, the + operator is available for concatenation. Adding a float f to a string s produces a string composed of the concatenation of the float value in decimal notation and the string s:

```
"hello" + " " + "world!";
[01544390:notag] "hello world!"
"number : "+6;
[01544390:notag] "number: 6.000000"
"number : "+string(6);
[01544390:notag] "number: 6"
```

The string function is explained below.

1.7.3 List

The standard + operator is available. It will concatenate lists or add elements to existing lists.

1.7.4 Binary

Binary can be concatenated using the + operator. The result is a binary containing the concatenation of both buffers. The parameter list is the one of the first binary if it has a list, the second otherwise. This functionality is mainly available to enable the aggregation of sound buffers.

This example uses the loop control structure that we will describe later (it is equivalent to while(true)), and the micro device, available for example on the Aibo robot. It stores 10 seconds of sound in a buffer, using the + operator:

sound = bin 0; +timeout(10s): loop sound = sound + micro;

1.7.5 Boolean

Booleans are treated like numerical expressions and can be used in tests for commands like if, while, or others. Standard operators like && || ! are available. Tests can be enclosed between parenthesis (). The true and false constants are available.

Expressions can be compared using the usual == != > < >= < comparison operators.

URBI defines also some fuzzy equality tests:

a ~= b: Means that $|a - b| \le global.epsilontilde$

global.epsilontilde is a global variable with default value equal to 1.0. This value can be modified to suit particular needs.

a %= b: Means that $|1 - a/b| \le global.epsilonpercent$

global.epsilonpercent is a global variable with default value equal to 0.05. This value can be modified to suit particular needs.

a = \sim = b: Means that |a-b| <= a - > delta + b - > delta. This fuzzy test uses the delta property of the expressions to compute the tolerance on the test. It is extremely useful to test motor values or any type of variable with a significant variance.

1.7.6 Standard functions

Included in the language is a set of standard functions, with their usual semantics:

sin asin cos acos tan atan exp log round random trunc sqr sqrt abs

There is also a string function, which takes a float and returns the integer part of the float converted into a string. This can be useful because the default conversion of a float keeps the decimal values, even if they are zeros.

Examples:

```
string(4+1);
[01544390:notag] "5"
sin(pi/2);
[01544390:notag] 1.000000
random(100); // random integer value between 0 and 100
[01544390:notag] 33.000000
```

Here are some useful functions that can be used with strings:

- strlen(s) : returns the length of the string s.
- strsub(s,pos,n) : returns the substring of s starting at position pos and containing n characters. Any overflow is truncated appropriately.

1.8 Control and event catching structures

The following control and event catching structures are available in URBI. Some are standard structures common in many known languages like C/C++, some are more specific to URBI. In the examples given, command1 or command2 designates a single command or a set of commands enclosed between brackets $\{ \ \}$.

The notion of *cycle* is used in the following descriptions. The URBI server is processing the command tree at specified intervals (32ms for Aibo robots). The complete examination of the command tree is called a cycle. At every cycle, an global.update event is emitted (see event emission in 1.9). The command noop, for example, does nothing but takes exactly one cycle to complete, which is equivalent to wait (global.update). This tool is useful to describe the semantic of the following commands.

1.8.1 Control structures

if (test) command1 else command2

This performs a standard if branching. Note that there is no terminating symbol between command1 and else. The *else* part is optional.

This command terminates when the selected command (1 or 2) terminates.

Example (see 1.13 for details about the echo command):

```
i=4;
if ( i > 3 ) echo "great" else echo "small";
[01544390:notag] *** great
```

while (test) command

This performs a standard while loop.

The loop terminates when command terminates and the test evaluation returns false. For the moment, there is at least one cycle between each execution of command (technically speaking: command is executed in parallel (&) with noop). So, it is possible to create an infinite loop but the server will never freeze. *This will change in future specifications and the noop inclusion might disappear*

Example:

```
i=0;
while (i<=2) {
    i:echo i;
    i++
};
[09696528:i] *** 0
[09696558:i] *** 1
[09696590:i] *** 2
```

Note how the timestamp is incremented at each turn.

while (test) command

This is similar to the while command except that each instance of command is connected with a | time operator. This is equivalent to:

```
command_instance1 | // perform test
command_instance2 | // perform test
...
```

The loop terminates when the last instance of command terminates and the test evaluation returns false.

Note that in the case of while |, it is possible to create an infinite loop that would freeze the server.

Example:

```
i=0;
while | (i<=2) {
    i:echo i;
    i++
};
[09696528:i] *** 0
[09696528:i] *** 1
[09696528:i] *** 2
```

Note how the timestamp is identical at each turn. The while | is almost the same than the while command except that since the | insures that all commands will be executed without a time gap, their is no cycle loop lost in the execution process. It can be useful to speed up the execution of a code, but it might also freeze the server.

for (instruction1; test; instruction2) command

This has the usual semantics of the C "for". instruction1 and instruction2 must be single instructions (no time operator or brackets). It is equivalent to:

```
instruction1;
while (test) {
   command |
   instruction2
}
```

for | (instruction1; test; instruction2) command

Just like the for command above, this is equivalent to:

```
instruction1;
while | (test) {
   command |
   instruction2
}
```

for & (instruction1; test; instruction2) command

This is similar to the for command except that each instance of command is connected with a & time operator. More precisely, this is equivalent to:

```
instruction1;
{ command_instance1 & { instruction2 | // perform test
{ command_instance2 & { instruction2 | // perform test
... }}}
```

The loop terminates when all the instances of command and instruction2 terminate and the test evaluation returns false.

Note that in the case of for &, it is possible to create an infinite loop that would freeze the server.

Example:

```
for & (i=0;i<=2;i++)
    i:echo i;
[09696528:i] *** 0
[09696528:i] *** 1
[09696528:i] *** 2</pre>
```

Note how the timestamp is identical at each turn. The value of i changes because the echo have virtually an execution time of zero seconds and so instruction2 (here "i++"), is executed right after the echo. This of course is the reason why the loop terminates at some point.

loop command

loop is equivalent to at (global.update). It loops and never terminates.

loopn (num) command

loopn (num) loops num times the command and terminates. It is equivalent to (but n is virtual here):

```
for (n=0; n<num; n++)
  command</pre>
```

loopn & (num) command

loopn & (num) loops num times the command in parallel, with the time operator &, and then terminates.

It is equivalent to (but n is virtual here):

```
for & (n=0; n<num; n++)
    command</pre>
```

loopn | (num) command

loopn | (num) loops *num* times the command in series, with the time operator |, and then terminates.

It is equivalent to (but n is virtual here):

```
for | (n=0; n<num; n++)
command</pre>
```

1.8.2 Soft tests

In the following, some event catching structures like at, whenever or wait can make use of *soft tests*. Those tests are called *soft tests* because they can integrate an hysteresis threshold, prefixed by the \sim separator. Consider the two following example of soft test:

(headSensor == 1 ~ 50ms) $\sim 50ms$ indicates that the test must be true during a least 50ms before the soft test turns to true.

These "soft tests" are very useful in robotics to conveniently set conditions that are fuzzy and more resistant to noise, and to avoid constant triggering on and off around the target value of the test.

1.8.3 Event catching structures

at (softtest) command1 onleave command2

The at command is an event catcher. This command never terminates and run in the background. At the moment when the test becomes true, command1 is executed once. Then, at the moment when the test becomes false again, command2 is executed once, if it is defined (the onleave part is optional). Then again, at waits for the test to become true to execute command1 once and so on.

command1 and command2 must terminate before the soft test is reexamined again. So if one of those commands does not terminate, the event catching capability will be lost.

The difference between at and if is that if does the test only once. If the test fails, it will never try again because the if command terminates. The at command never terminates and performs the test at each cycle. It will also require that the test becomes false before the command can be executed again.

at & (softtest) command1 onleave command2

The at & command is similar to the at command except that command1 and command2 are executed in parallel with the at command when they are executed. In other words, the softtest is reexamined at each cycle, even if the commands are not terminated yet. For example, there can be several instances of command1 running at the same time if the test switches between false and true several times during the execution time of the first command1.

If one of the commands does not terminate, the event catching capability will not be lost.

whenever (softtest) command1 else command2

The whenever command is also an event catcher. This command never terminates and run in the background. Whenever the softtest is true, command1 is executed, otherwise, if it is given, command2 is executed (the else part is optional). command1 or command2 must terminate before the soft test is reexamined again. So if one of those commands does not terminate, the event catching capability will be lost.

whenever is different from at because command1 will be executed each time the test is true, and not only at the time it *becomes* true. In that sense, whenever is close to while, the difference being that when the test fails it does not terminate but waits for the test to become true again.

wait n

wait n waits n milliseconds and terminates. This is useful to make a pause or control the execution time or synchronization of different commands.

waituntil (softtest)

waituntil (softtest) waits until softtest becomes true and terminates. This is useful to make a pause or control the execution time or synchronization of different commands.

timeout (time) command

timeout (time) command runs command and stops it when the specified time is over. If the command ends before the specified time, timeout has no effect. time is expressed in milliseconds or with a time expression like 12m54s.

```
stopif (softtest) command
```

stopif (softtest) command runs command and stops it if the specified softtest becomes true. If the command ends before the specified softtest is true, stopif has no effect.

```
freezeif (softtest) command
```

freezeif (softtest) command runs command and freezes it if the specified softtest becomes true. When it turns false again, the command is unfreezed. This behavior ends when the command terminates.

1.9 Emit events

Event programming is a fundamental feature and a good way of doing asynchronous programming. The basic idea of event programming is that some command emit an event and some other catches this event and do something.

1.9.1 Simple events

To emit an event, the emit command is available, and you can use at or whenever to catch it:

```
at (myevent) echo "event triggered!";
emit myevent;
[00143009:notag] *** event triggered!
```

Note that the myevent event here is local to the connection. If you want to make the event visible from other connection, you should use a prefix, like myprefix.myevent.

1.9.2 Events with parameters

You can add parameters to events like this:

```
emit myevent(1, "hello");
```

The parameters can be retrieved when the event is catched:

```
at (myevent(x,y))
    echo "catch two: " + x + " " + y;
at (myevent(1,x))
    echo "catch one: " + x;
```

The second at here is doing a filtering on the event parameters, accepting only events whose first parameter equals 1:

```
emit myevent(1,"hello");
[00143009:notag] *** catch two: 1.000000 hello
[00143009:notag] *** catch one: hello
emit myevent(2,15);
[00143009:notag] *** catch two: 2.000000 15.000000
```

1.9.3 Event duration

Events usually have a virtually null duration, they are just spikes (Dirac functions). You can explicitly requests that an event lasts for a certain duration by specifying this duration between parenthesis like this:

```
emit(10s) myevent;
emit(15h12m) myevent(1,"hello");
```

This will induce a difference between at and whenever event catcher.

1.9.4 Pulsing events: the every **command**

You can have any command repeated at specific time intervals in URBI, using the every command. The following example will say "hello" every 10 minutes:

every (10m) echo "hello";

You can use the every command to pulse events at regular intervals:

every (100ms) emit pulse;

To stop the emission, just use stop with the appropriate tag (see 1.13.3):

mypulse:every (100ms) emit pulse; stop mypulse;

1.9.5 Predefined pulsing events

If the URBI server is implemented in a cycle, like on the Aibo, the global event global.update will pulse at every loop. You can use it as a beat for your programs.

Specific events related to devices must also be available. In general, device.update signals that the value of the device device has been changed.

1.10 Grouping

An important feature of URBI is the capacity to group devices. This is done with the group command:

```
group virtualdevice { device1, device2, ... }
```

Example:

```
group legLF {legLF1, legLF2, legLF3};
group legs {legLF, legLH, legRF, legRH};
```

This grouping feature is useful when a *method* is called on a device (or a virtual device), or to make a *multi-device assignment*: the command is executed for this device and then it is recursively passed to any child subdevice. In other terms, using the example above, the command legLF.val = 0 will set the value of lefLF1, legLF2 and legLF3 to 0. This child passing mechanism differs from the usual object oriented inheritance mechanism because object oriented hierarchies are based on "is-a-kind-of" relationships whereas the grouping hierarchies are based on "is-a-part-of" relationships.

NB: It is possible to block the passing mechanism by prefixing the device name with the word only. In that case, only the specified device will execute the command.

Example:

```
group ab {a,b};
ab.n = 5;
a:a.n, b:b.n, ab:ab.n;
[09696528:a] 5.000000
[09696528:b] 5.000000
[09696528:ab] 5.000000
a.n = 1;
b.n = 2;
only ab.n = 5;
a:a.n, b:b.n, ab:ab.n;
[09696528:a] 1.000000
[09696528:b] 2.000000
[09696528:ab] 5.000000
```

For any robot, it is advised to design a hierarchy of devices, with *robot* on the top. This is usually done via a URBI.INI file (see 1.12).

Device grouping will also affect the way events are processed. If a is a virtual device containing b, every b.x event will also trigger an a . x event.

1.11 Function definition

Functions can be defined in URBI using the def keyword, followed by the function name in prefix.suffix notation or simply suffix for a function local the connection and the parameters between brackets (or an open/close bracket if there is no parameters). The parameters are always local to the function call. Non-global (i.e. without prefix) variable in the function body are also local to the function call.

Example:

```
def adding(x,y)
{
    z = x+y;
    return z
};

def print(x)
{
    echo x;
    if (x<0) return</pre>
```

```
else
    echo sqrt(x)
}
```

Functions can return values or void. If they return values, they can be used in expressions.

Recursive calls of functions are allowed in URBI, since all local variables and parameters are local to the function call (the server gives them a temporary prefix which is linked to the function call).

1.12 Files

If the robot has a filesystem available, it is possible to do some very simple file manipulation.

load(file)

This command loads the file with name file given as a string, parse its content and execute the commands, as if they had been entered by hand at the position of the "load".

save(file,s)

This command creates a file with name file (file is a string) and stores the string s in it. The file can be reloaded later with the load command.

URBI.INI

The URBI.INI file, located in the root of the filesystem, is executed when the server starts. It is typically used to execute some initialization commands specific to the robot, like playing a welcome sound.

The commands are executed in a virtual connection called the "ghost connection". Any output from this connection is sent to the debug output, which is most of the case port 59000.

Standard usage is to load some custom files at the beginning of the URBI.INI file, to activate "modules". The most usual module is "std.u" which contain standard groups and functions useful for your robot.

CLIENT.INI

The CLIENT.INI file, located in the root of the filesystem, is executed when a new client starts. It runs in the new client's connection and any output will be displayed to this connection.

1.13 Standard commands, operators or global variables

1.13.1 echo

The echo command displays a value in a system message. Used with a binary value, it only displays the header containing the bin size and the parameters. This is useful to control the content of a binary without viewing the binary buffer itself.

The echo command can also be used to display system messages from one connection to the other. Simply use the syntax: echo <connectionTag> : <value>

Examples:

```
echo 45;
[01544390:notag] *** 45
echo "hello";
[01544390:notag] *** hello
echo U597767000 : "hello"
/* equivalent to echo "hello" on
    connection U597767000
*/
echo camera;
[01544390:notag] *** BIN 9437 jpeg 208 160
```

You can choose to display the result of the echo command in another connection by adding the connection modifier, followed by the connection ID or "all" for all connection, or "other" for all connections except the current one:

```
echo "hello" connection:"U452145754";
echo "hello" connection:"all";
echo "hello" connection:"other";
```

1.13.2 exec

exec(s) parses the string s and run the corresponding command, as if it had been entered by hand.

Examples:

```
exec( "a=1;" );
a;
[01544390:notaq] 1.000000
```

1.13.3 stop, block, unblock, freeze, unfreeze

These are fundamental and very useful commands to control the execution of programs:

stop thetag will stop any command whose tag is thetag.

block thetag will block any command whose tag is thetag, preventing it to be executed.

unblock thetag will unblock the commands whose tag is thetag, previously blocked by a block command.

freeze thetag will freeze any command whose tag is thetag, putting it in a "suspended" mode unfreeze thetag will unfreeze the commands whose tag is thetag, resuming it where it was when it was freezed.

1.13.4 ping

The ping command sends the following system message:

*** pong time=t

t is the server uptime in milliseconds, with microseconds expressed as decimals. This is the real server time at the moment when the message is sent, it may and will differ from the timestamp in the message header and can be used to do some accurate time measures on the server.

1.13.5 motor on, motor off, motor switch

The motor on command starts the motors (the joints or wheels become rigid). The motor off does the opposite. Default status at start is "motor off". Using motor on usually increases power consumption.

motor switch will switch from on to off or from off to on.

1.13.6 connections, vars, uservars

connections lists the currently opened connections. The "Ghost Connection" is the connection running URBI.INI.

vars show all variables

uservars show all user defined variables

1.13.7 stopall, undefall, killall, disconnect

The commands killall and disconnect are followed by a connection identificator. Each client has a unique connection identificator, displayed at start in the URBI header and accessible later through the connectionID variable.

killall id will empty the command tree of the connection whose ID is *id*. Every command will be stopped.

undefall id will clear all function and variable definitions.

disconnect id will close the connection whose ID is *id*.

stopall is equivalent to killall on every connection opened. This is a powerful command that stops everything running on the robot and should be used carefully.

1.13.8 quit, reboot, shutdown, reset

quit closes the current connection.

reboot reboots the robot and shutdown switches the robot off.

reset reset the robot by doing a "software" reboot: all functions are killes, all variables are lost and URBI.INI and CLIENT.INI are reloaded.

1.13.9 Useful functions that are part of the specification

The following functions are part of the URBI specification:

- freemem() : returns the amount of free memory in bytes.
- power(): returns the remaining battery power, expressed as a float between 0 and 1 (0:empty, 1:full).
- time() : returns the server uptime, in milliseconds, with a microsecond precision in the decimals if it is available for the robot.

- cpuload() : returns the load of the robot CPU: 0 for idle and 1 for full capacity. Going above 1 will degrade the quality of the executed commands and time constraints are not guaranteed anymore.
- loadwav("file.wav") : this will return a binary equal to the wav file specified. Useful to store in the URBI memory some frequently used sound files.

1.14 URBI Header

When a new client starts, it must receive a standard URBI header for the server to be URBI compliant.

Here is an example of a URBI Header coming from the 1.0 version of URBI server for Aibo robot, as distributed on urbiforge (the number before the brackets is not part of the header, it is here just to simplify the line reference in the following):

```
2 [00139464:start] URBI Language specif 1.3
                                     _
                                        Copyright (C) 2005 JC Baillie
3 [00139464:start] URBI Kernel version 1.0 build 4124
4 [00139464:start]
                 URBI Server version 1.0 for Aibo ERS2xx/ERS7 Robots
5 [00139464:start]
                 (C) 2004-2005 Jean-Christophe Baillie
6 [00139464:start]
7 [00139464:start]
8 [00139464:start] URBI comes with ABSOLUTELY NO WARRANTY;
9 [00139464:start] This software is free, and you are welcome to use
               it under certain conditions; see LICENSE for details.
10[00139464:start]
11[00139464:start]
12[00139464:start]
               See http://www.urbiforge.com for news and updates.
14[00139464:ident] ID: U597766392
```

All lines between line 1 and 13 (included) must have the start tag.

Line 2 is mandatory, specifying the corresponding version of the URBI Specification that the server implements and the copyright holder for the specification. The "build" indication is optional and indicates a reference on the build history. Two version with the same build number are exactly the same.

Line 3 is mandatory for URBI Servers who derive from the urbiforge implementation, to make a clear reference to the kernel version used. The source code of URBI is divided into a kernel part¹ and a robot-specific part. Each of these parts can have its own version number. Lines 5 and 6 can be used to specify the version of the robot-specific part and the type of robot supported.

Lines 1, 4, 7, 11 and 12 are optional. They are here just to ease the reading of the header.

Line 8 is mandatory.

Line 9 and 10 are mandatory to indicates where the license can be found.

Line 12 is mandatory for any server with code that derives from the URBI kernel on urbiforge.

Line 14 is mandatory, with a ident tag. It specifies the tag of the connection (to be used for local variables prefix or killall, disconnect commands).

¹The kernel contains all the code that is not specific to a robot (the majority of the code). The robot-specific part derives some robot-specific instances from the kernel object and instantiates some virtual members

Chapter 2

URBI grammar

2.1 Bison grammar

Here is the full URBI grammar which is part of the specification. Terminal symbols are written in upper case and are described in the following section.

```
0 $accept: ROOT $end
 1 ROOT: refvariable ASSIGN binary SEMICOLON
 2
       | taggedcommands
 3 taggedcommands: taggedcommand
                 | taggedcommands COMMA taggedcommands
 4
 5
                 | taggedcommands SEMICOLON taggedcommands
 6
                 taggedcommands PIPE taggedcommands
 7
                 | taggedcommands AND taggedcommands
 8 taggedcommand: command
 9
                | IDENTIFIER flags COLON command
10
                  IDENTIFIER COLON command
11
                | flags COLON command
12 flags: FLAG
      | FLAG flags
13
14
        | FLAGTIME LPAREN expr RPAREN flags
15
        FLAGTIME LPAREN expr RPAREN
        | FLAGID LPAREN expr RPAREN
16
17
        | FLAGID LPAREN expr RPAREN flags
18
          FLAGTEST LPAREN softtest RPAREN flags
        | FLAGTEST LPAREN softtest RPAREN
19
20 command: instruction
21
          | LBRACKET taggedcommands RBRACKET
22 instruction: /* vide */
23
              | NOOP
```

24	refvariable ASSIGN expr namedparameters
25	DEF refvariable ASSIGN expr namedparameters
26	property ASSIGN expr
27	expr
28	IDENTIFIER SWITCH
29	RETURN
30	RETURN expr
31	ECHO expr namedparameters
32	GROUP IDENTIFIER LBRACKET identifiers RBRACKET
33	GROUP IDENTIFIER
34	ALIAS variable variable
35	ALIAS
36	OPERATOR
37	OPERATOR ID IDENTIFIER
38	OPERATOR_ID_IDENTIFIER OPERATOR_VAR_variable
39	WAIT expr
40	EMIT purevariable
41	EMIT purevariable LPAREN parameterlist RPAREN
42	EMIT LPAREN expr RPAREN purevariable
43	EMIT LPAREN expr RPAREN purevariable LPAREN parameterlist
	RPAREN
44	WAITUNTIL softtest
45	refvariable MINUSMINUS
46	refvariable PLUSPLUS
47	DEF
48	DEF refvariable
49	DEF LBRACKET refvariables RBRACKET
50	DEF IDENTIFIER LBRACKET identifiers_semicolon RBRACKET
51 @1: /* vide	*/
52 instruction.	DEF variable LPAREN identifiers RPAREN @1 taggedcommand
53	IF LPAREN expr RPAREN taggedcommand
54	IF LPAREN expr RPAREN taggedcommand ELSE taggedcommand
55	EVERY LPAREN expr RPAREN taggedcommand
56	AT LPAREN softtest RPAREN taggedcommand
57	AT LPAREN softtest RPAREN taggedcommand ONLEAVE taggedcommand
58	AT AND LPAREN softtest RPAREN taggedcommand
59	AT AND LPAREN softtest RPAREN taggedcommand ONLEAVE
1	taggedcommand
60	WHILE LPAREN expr RPAREN taggedcommand
61	WHILE PIPE LPAREN expr RPAREN taggedcommand
62	WHENEVER LPAREN softtest RPAREN taggedcommand
63	WHENEVER LPAREN softtest RPAREN taggedcommand ELSE
	taggedcommand
64	LOOP taggedcommand
65	FOREACH purevariable IN expr LBRACKET taggedcommands RBRACKET
66	FOREACH AND purevariable IN expr LBRACKET taggedcommands
	RBRACKET
67	FOREACH PIPE purevariable IN expr LBRACKET taggedcommands
	RBRACKET
68	LOOPN LPAREN expr RPAREN taggedcommand

```
69
               | LOOPN PIPE LPAREN expr RPAREN taggedcommand
 70
               | LOOPN AND LPAREN expr RPAREN taggedcommand
              | FOR LPAREN instruction SEMICOLON expr SEMICOLON instruction
 71
                RPAREN taggedcommand
               | FOR PIPE LPAREN instruction SEMICOLON expr SEMICOLON instruction
 72
                RPAREN taggedcommand
 73
               | FOR AND LPAREN instruction SEMICOLON expr SEMICOLON instruction
                RPAREN taggedcommand
 74 array: /* vide */
        | LSBRACKET expr RSBRACKET array
 75
 76 purevariable: DOLLAR LPAREN expr RPAREN
 77
               | IDENTIFIER array
 78
               | STRUCT array
 79 variable: purevariable
        | COLON purevariable
 80
 81
           | purevariable NORM
 82
           | purevariable VARERROR
           | purevariable DERIV
 83
 84
           purevariable DERIV2
           | purevariable TRUEDERIV
 85
           | purevariable TRUEDERIV2
 86
 87 refvariable: variable
              | AROBASE variable
 88
 89 property: purevariable DIR IDENTIFIER
 90 namedparameters: /* vide */
                  | IDENTIFIER COLON expr namedparameters
 91
 92 binary: BIN NUM
        BIN NUM rawparameters
 93
 94 timeexpr: TIMEVALUE
 95
        | timeexpr TIMEVALUE
96 expr: NUM
 97 timeexpr
       STRING
98
99
       | LSBRACKET parameterlist RSBRACKET
100
       | PERCENT variable
101
        property
       | refvariable LPAREN parameterlist RPAREN
102
      variable
103
104
      | expr PLUS expr
105
      | expr MINUS expr
106
      | expr MULT expr
       | expr DIV expr
107
108
      | COPY expr
```

```
109
        | MINUS expr
110
         expr EXP expr
111
        | LPAREN expr RPAREN
112
        TRUECONST
        | FALSECONST
113
114
        | expr EQ expr
115
        expr REQ expr
116
        expr DEQ expr
117
        | expr PEQ expr
118
        | expr NE expr
119
        expr GT expr
120
        | expr GE expr
121
        expr LT expr
122
        | expr LE expr
123
        BANG expr
124
        expr ANDOPERATOR expr
125
        expr OROPERATOR expr
126 parameterlist: /* vide */
127
                | parameters
128 parameters: expr
129
       expr COMMA parameters
130 rawparameters: NUM
131
                IDENTIFIER
132
                | NUM rawparameters
133
                 | IDENTIFIER rawparameters
134 softtest: expr
           | expr TILDE expr
135
136 identifiers: /* vide */
137
              IDENTIFIER
138
               | IDENTIFIER COMMA identifiers
139 identifiers semicolon: /* vide */
140
                          IDENTIFIER
141
                          IDENTIFIER SEMICOLON identifiers semicolon
142 refvariables: /* vide */
143
               | refvariable
144
                | refvariable SEMICOLON refvariables
```

2.2 Terminal symbols and flex regexp

The following are the terminal symbols definitions, using regexp in the style of a flex definition file.

NUMBER ((. [0-9] +)) ([0-9] + (. [0-9] *) ?)

ID [a-zA-Z_] [_A-Za-z0-9] * STRING \"([^"]|"\\\"")*\"

Comment definition:

"//"[^\n] * "#"[^\n] * \/*(.|"\n") **\/

Other terminal symbols:

```
"inf"
           : NUM (infinite value)
"pi"
           : NUM (value of pi)
"+report" : FLAG
"+error"
           : FLAG
"+end"
            : FLAG
"+begin"
           : FLAG
"+timeout"
           : FLAGTIME
"+stop"
           : FLAGTEST
            : FLAGTEST
"+freeze"
{NUMBER}"s" : TIMEVALUE
{NUMBER}"ms" : TIMEVALUE
{NUMBER}
            : NUM
{STRING}
            : STRING
stop
unblock
block
unfreeze
freeze
killall
disconnect : OPERATOR_ID
info
          undef
             : OPERATOR VAR
mix
normal
cancel
discard |
queue
             : STRING
on
      : SWITCH
off
defcheckon |
defcheckoff
ping
```

devices			
vars			
uservars	Ì		
commands	i		
debugon	i		
debugoff	ł		
quit	ł		
stopall			
reboot			
	I		
shutdown		:	OPERATOR
if		:	IF
else			ELSE
while			WHILE
at		:	AT
			EVERY
every			
whenever		:	WHENEVER
onleave			ONLEAVE
for			FOR
loop			LOOP
loopn			LOOPN
group		:	GROUP
echo		:	ECHO
wait		:	WAIT
def		:	DEF
return		:	RETURN
"BIN"		:	BIN
bin		:	BIN
noop		:	NOOP
true		:	TRUECONST
false		:	FALSECONST
booleval		:	TEST
сору		:	COPY
alias		:	ALIAS
"'n"		:	NORM
"'e"		:	VARERROR
"'d"		:	TRUEDERIV
"′dd"		:	TRUEDERIV2
			DERIV2
" / "		•	DERIV
"&&"			ANDOPERATOR
" "			OROPERATOR
";"			SEMICOLON
":"			COLON
","		:	COMMA
"&"		:	AND
" "		:	PIPE
"!"			BANG
""		:	MINUSMINUS
"++"		:	PLUSPLUS
"*"		:	MULT

"/"	: DIV
"+"	: PLUS
"^"	: EXP
"_"	: MINUS
"="	: ASSIGN
"=="	: EQ
"~="	: REQ
"=~="	: DEQ
"%="	: PEQ
"!="	: NE
">"	: GT
">="	: GE
"<"	: LT
" <= "	: LE
"~"	: TILDE
"응"	: PERCENT
"@"	: AROBASE
" ("	: LPAREN
")"	: RPAREN
"["	: LSBRACKET
"]"	: RSBRACKET
" { "	: LBRACKET
" } "	: RBRACKET
"\$"	: DOLLAR
{ID}	: IDENTIFIER
{ID}"."{ID}	: STRUCT
$[\t n\r] +$	/* eat up whitespace */